

Mapping properties via Mathematica


With the aid of Mathematica, one can better visualize mapping properties of a complex function. Let us consider the *exponential function* for example.

We know that the $f(z) = e^z$ is periodic with period $2\pi i$. To see all possible values of $f(z)$ (or in other words, the range of f), it suffices to restrict the domain of f to an infinite horizontal strip of width 2π , for instance $(-\infty, \infty) \times (-\pi, \pi]$, instead of the whole complex plane. Recall that f can be seen as a geometric transformation on the complex plane. To view the image of the rectangle $[-5, 5] \times [-\pi, \pi]$ under f , use the **ParametricPlot** command. This command has several different syntaxes. The one that we need is essentially the following:

```
ParametricPlot[{g[x,y],h[x,y]}, {x, a, b}, {y, c, d}]
```


where $g(x, y)$ and $h(x, y)$ are two scalar-valued functions. The above command draws the set of all points of coordinate $(g(x, y), h(x, y))$ where (x, y) runs all over the rectangle $[a, b] \times [c, d]$. For example,

```
ParametricPlot[{x*Sin[y], y*Sin[x]}, {x, -2, 2}, {y, -2, 2}]
```



To draw the image of a rectangle $[-5, 5] \times [-\pi, \pi]$ under the function e^z , use the command

```
ParametricPlot[{Re[Exp[x + y*I]], Im[Exp[x + y*I]]}, {x, -5, 5}, {y, -Pi, Pi},  
PlotRange -> Full]
```




This command can be shortened as follows:

```
ParametricPlot[ReIm[Exp[x + y*I]], {x, -5, 5}, {y, -Pi, Pi}, PlotRange -> Full]
```

To see how each grid line $x = a$ and $y = b$ is transformed into, one simply adds the option **Mesh** and **MeshStyle** to the above command.

```
ParametricPlot[ReIm[Exp[x + y*I]], {x, -5, 5}, {y, -Pi, Pi},  
PlotRange -> Full, Mesh -> Automatic, MeshStyle -> {Red, Green}]
```




The grid lines $x = a$ are transformed into red curves (circles), while the grid lines $y = b$ are transformed into green curves (ray emitting from the origin). It would be helpful to see those transformations in motion. To draw the rectangles $[-s, s] \times [-\pi, \pi]$ side by side with their images as s runs from 0.1 to 10, use the commands:

```
p[s_] := ParametricPlot[ReIm[x + y*I], {x, -s, s}, {y, -Pi, Pi},
  PlotRange -> Full, MeshStyle -> {Red, Green}, Mesh -> Automatic]

q[s_] := ParametricPlot[ReIm[Exp[x + y*I]], {x, -s, s}, {y, -Pi, Pi},
  PlotRange -> Full, MeshStyle -> {Red, Green}, Mesh -> Automatic]

Manipulate[{p[s], q[s]}, {s, 0.1, 10}]
```




Moving the cursor to vary s , one can see that the image of each rectangle is an annulus. The outer radius grows larger while the inner radius shrinks as $s \rightarrow \infty$. A good guess is that the image of the infinite strip $(-\infty, \infty) \times (-\pi, \pi]$ under the exponential function is $\mathbb{C} \setminus \{0\}$.

Another nice feature of the **ParametricPlot** function is that it not only shows image (under e^z) of each point on the left graph, but also indicates if any point on the right graph comes from multiple points on the left graph. The same blue background on the right as on the left graph indicates that each point on the right comes from exactly one point on the left. Let us the range of y to $\{y, -\text{Pi}, s\}$ and let s vary from $\pi - 1$ to $\pi + 1$:


```
p1[s_] := ParametricPlot[ReIm[x + y*I], {x, -1, 1}, {y, -Pi, s},
  PlotRange -> Full, MeshStyle -> {Red, Green}, Mesh -> Automatic]

q1[s_] := ParametricPlot[ReIm[Exp[x + y*I]], {x, -1, 1}, {y, -Pi, s},
  PlotRange -> Full, MeshStyle -> {Red, Green}, Mesh -> Automatic]

Manipulate[{p1[s], q1[s]}, {s, Pi-1, Pi+1}]
```



When $s < \pi$, the image under f of the rectangle $[-1, 1] \times [-\pi, s]$ is not yet a full annulus. When $s > \pi$, there occurs a darker blue region on the right graph, suggesting that each point in this region comes from multiple points on the left. Thus, it is reasonable to conclude that the exponential



function maps

- the strip $(-\infty, \infty) \times (-\pi, \pi]$ one-to-one and onto (in another word, bijectively) the punctured complex plane $\mathbb{C} \setminus \{0\}$,
- the open strip $(-\infty, \infty) \times (-\pi, \pi)$ one-to-one and onto the slit complex plane $\mathbb{C} \setminus \mathbb{R}_{\leq 0}$.

A careful double-check by calculation confirms these conclusions. However, detail computation is somewhat complicated, which is not of our interest here.